

Mechanical Theorem Proving in Tarski's geometry.

Julien Narboux

► **To cite this version:**

Julien Narboux. Mechanical Theorem Proving in Tarski's geometry.. Automated Deduction in Geometry 2006, Francisco Botana, Aug 2006, Pontevedra, Spain. pp.139-156, 10.1007/978-3-540-77356-6 . inria-00118812

HAL Id: inria-00118812

<https://hal.inria.fr/inria-00118812>

Submitted on 6 Dec 2006

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

Mechanical Theorem Proving in Tarski's Geometry.

Julien Narboux

Équipe LogiCal, LIX

École Polytechnique, 91128 Palaiseau Cedex, France

Julien.Narboux@inria.fr,

<http://www.lix.polytechnique.fr/Labo/Julien.Narboux/>

Abstract. This paper describes the mechanization of the proofs of the first height chapters of Schwabäuser, Szmielew and Tarski's book: *Meta-mathematische Methoden in der Geometrie*. The goal of this development is to provide foundations for other formalizations of geometry and implementations of decision procedures. We compare the mechanized proofs with the informal proofs. We also compare this piece of formalization with the previous work done about Hilbert's *Grundlagen der Geometrie*. We analyze the differences between the two axiom systems from the formalization point of view.

1 Introduction

Euclid is considered as the pioneer of the axiomatic method, in the *Elements*, starting from a small number of self-evident truths, called postulates, or common notions, he derives by purely logical rules most of the geometrical facts that were discovered in the two or three centuries before him. But upon a closer reading of Euclid's *Elements*, we find that he does not adhere as strictly as he should to the axiomatic method. Indeed, at some steps in certain proofs he uses a method of "superposition of triangles" and this kind of justifications can not be derived from his set of postulates.

In 1899, in *der Grundlagen der Geometrie*, Hilbert proposed a new axiom system to fill the gaps in Euclid's system.

Recently, the task consisting in mechanizing Hilbert's *Grundlagen der Geometrie* has been partially achieved. A first formalization using the Coq proof assistant [Coq04] was proposed by Christophe Dehlinger, Jean-François Dufourd and Pascal Schreck [DDS00]. This first approach was realized in an intuitionist setting, and concluded that the decidability of point equality and collinearity is necessary to perform Hilbert's proofs. Another formalization using the Isabelle/Isar proof assistant [Pau] was performed by Jacques Fleuriot and Laura Meikle [MF03]. These formalizations have concluded that Hilbert proofs are in fact not fully formal¹, in particular degenerated cases are often implicit in the

¹ Note that in the different editions of *die Grundlagen der Geometrie* the axioms were changed, but the proofs were not always changed accordingly.

presentation of Hilbert. The proofs can be made more rigorous by machine assistance.

In the early 60s, Wanda Szmielew and Alfred Tarski started the project of a treaty about the foundations of geometry based on another axiom system for geometry designed by Tarski in the 20s². A systematic development of euclidean geometry was supposed to constitute the first part but the early death of Wanda Szmielew put an end to this project. Finally, Wolfram Schwabhäuser continued the project of Wanda Szmielew and Alfred Tarski. He published the treaty in 1983 in German: *Metamathematische Methoden in der Geometrie* [SST83]. In [Qua89], Art Quaife uses a general purpose theorem prover to automate the proof of some lemmas in Tarski's geometry. In this paper we describe our formalization of the first eight chapters of the book of Wolfram Schwabhäuser, Wanda Szmielew and Alfred Tarski in the Coq proof assistant.

We will first describe the different axioms of Tarski's geometry and give an history of the different versions of this axiom system. Then we present our formalization of the axiom system and the mechanization of one example theorem. Finally we compare our formalization with existing ones and compare Tarski's axiomatic system with Hilbert's system from the mechanization point of view.

2 Motivations

We aim at two applications: the first one is the use of a proof assistant in the education to teach geometry [Nar05], the second one is the proof of programs in the field computational geometry.

These two themes have already been addressed by the community. Frédérique Guilhot has realized a large Coq development about euclidean geometry as it taught in french highschool [Gui05]. Concerning the proof of programs in the field of computational geometry we can cite the formalization of convex hulls algorithms by David Pichardie and Yves Bertot in Coq [PB01] and by Laura Meikle and Jacques Fleuriot in Isabelle [MF05]. In [Nar04], we have presented the formalization and implementation in the Coq proof assistant of the area decision procedure of Chou, Gao and Zhang [CGZ94].

Formalizing geometry in a proof assistant has not only the advantage of providing a very high level of confidence in the proof generated, it also permits to combine proofs about geometry with other kind of proofs such as the proof of the correctness of a program for instance. The goal which consist in using the same formal development about geometry for different purposes can only be achieved if we use the same axiomatic system. This is not the case for the time being.

The goal of our mechanization is to do a first step in this direction. We aim at providing very clear foundations for other formalizations of geometry and implementations of decision procedures.

² These historical pieces of information are taken from the introduction of the publication by Givant in 1999 [TG99] of a letter from Tarski to Schwabhäuser (1978).

Compared to Frédérique Guilhot formalization [Gui05], our development should be considered low level. Our formalization has the advantage of being based on the axiom system of Tarski which is of an extreme simplicity: two predicates and eleven axioms. But this simplicity has a price, our formalization is not adapted to the context of education. Indeed, some intuitively simple properties are hard to prove in this context. For instance, the proof of the existence of the midpoint of segment is obtained only at the end of the eighth chapter after about 150 lemmas and 4000 lines of proof. The small number of axioms impose a scheduling of the lemmas which is not always intuitive (some simple properties can only be proved late in the development).

3 Tarski's axiom system

Alfred Tarski worked on the axiomatization and meta-mathematics of euclidean geometry from 1926 until his death in 1983. Several axiom systems were produced by Tarski and his students. In this section, we first give an informal description of the propositions which appeared in the different versions of Tarski's axiom system, then we provide an history of these versions and finally we present the version we have formalized.

The axioms are based on first order logic and two predicates: *betweenness* and *equidistance* (or *congruence*). The ternary *betweenness* predicate βABC informally states that B lies on the line AC between A and C . The quaternary *equidistance* predicate $AB \equiv CD$ informally means that the distance from A to B is equal to the distance from C to D . In Tarski's geometry, only a set of points is assumed. In particular, lines are *defined* by two distinct points³.

3.1 Axioms

We reproduce here the list of propositions which appear in the different versions of Tarski's axiom system. We adopt the same numbering as in [TG99]. Free variables are considered to be implicitly quantified universally.

1 Reflexivity for equidistance

$$AB \equiv BA$$

2 Pseudo-transitivity for equidistance

$$AB \equiv PQ \wedge AB \equiv RS \Rightarrow PQ \equiv RS$$

3 Identity for equidistance

$$AB \equiv CC \Rightarrow A = B$$

³ In Hilbert's axiom system lines and planes are not *defined* but *assumed*.

4

4 Segment construction

$$\exists X, \beta Q A X \wedge AX \equiv BC$$

The segment construction axiom states that one can build a point on a ray at a given distance.




Fig. 1. Segment construction

5 Five segments

$$\begin{aligned} A \neq B \wedge \beta ABC \wedge \beta A'B'C' \wedge \\ AB \equiv A'B' \wedge BC \equiv B'C' \wedge AD \equiv A'D' \wedge BD \equiv B'D' \end{aligned} \Rightarrow CD \equiv C'D'$$

5₁ Five segments (variant)

$$\begin{aligned} A \neq B \wedge B \neq C \wedge \beta ABC \wedge \beta A'B'C' \wedge \\ AB \equiv A'B' \wedge BC \equiv B'C' \wedge AD \equiv A'D' \wedge BD \equiv B'D' \end{aligned} \Rightarrow CD \equiv C'D'$$

This second version differs from the first one only by the condition $B \neq C$.

6 Identity for betweenness

$$\beta ABA \Rightarrow A = B$$

The original Pasch axiom states that if a line intersects one side of a triangle and misses the three vertexes, then it must intersect one of the other two sides.

7 Pasch (inner form)

$$\beta APC \wedge \beta BQC \Rightarrow \exists X, \beta P X B \wedge \beta Q X A$$

7₁ Pasch (outer form)

$$\beta APC \wedge \beta QCB \Rightarrow \exists X, \beta AXQ \wedge \beta BPX$$




Fig. 2. Axioms of Pasch

7₂ Pasch (outer form) (variant)

$$\beta APC \wedge \beta QCB \Rightarrow \exists X, \beta AXQ \wedge \beta XPB$$

7₃ weak Pasch

$$\beta ATD \wedge \beta BDC \Rightarrow \exists X, Y, \beta AXB \wedge \beta AYC \wedge \beta YTX$$

Dimension axioms provide upper and lower bound for the dimension of the space. Note that lower bound axioms for dimension n are the negation of upper bound axioms for the dimension $n - 1$.

8(2) Dimension, lower bound 2

$$\exists ABC, \neg\beta ABC \wedge \neg\beta BCA \wedge \neg\beta CAB$$

There are three non collinear points.

8(n) Dimension, upper bound n

$$\exists ABCP_1P_2 \dots P_{n-1}, \bigwedge_{1 \leq i < j < n} P_i \neq P_j \wedge \bigwedge_{i=2}^{n-1} AP_1 \equiv AP_i \wedge BP_1 \equiv BP_i \wedge CP_1 \equiv CP_i \wedge \neg\beta ABC \wedge \neg\beta BCA \wedge \neg\beta CAB$$

9(1) Dimension, upper bound 1

$$\beta ABC \vee \beta BCA \vee \beta CAB$$

Three points are always on the same line.

9(n) Dimension, upper bound n

$$\bigwedge_{1 \leq i < j \leq n} P_i \neq P_j \wedge \bigwedge_{i=2}^n AP_1 \equiv AP_i \wedge BP_1 \equiv BP_i \wedge CP_1 \equiv CP_i \Rightarrow \beta ABC \vee \beta BCA \vee \beta CAB$$

6

9₁(2) Dimension, upper bound 2 (variant)⁴

$$\exists Y, (ColXYA \wedge \beta BYC) \vee (ColXYB \wedge \beta CYA) \vee (ColXYC \wedge \beta AYB)$$

10 Euclid's axiom

$$\beta ADT \wedge \beta BDC \wedge A \neq D \Rightarrow \exists X, Y \beta ABX \wedge \beta ACY \wedge \beta XTY$$

10₁ Euclid's axiom (variant)

$$\beta ADT \wedge \beta BDC \wedge A \neq D \Rightarrow \exists X, Y \beta ABX \wedge \beta ACY \wedge \beta YTX$$

11 Continuity

$$\exists a, \forall xy, (x \in X \wedge y \in Y \Rightarrow \beta axy) \Rightarrow \exists b, \forall xy, x \in X \wedge y \in Y \Rightarrow \beta xby$$

Schema 11 Elementary Continuity (schema)

$$\exists a, \forall xy, (\alpha \wedge \beta \Rightarrow \beta axy) \Rightarrow \exists b, \forall xy, \alpha \wedge \beta \Rightarrow \beta xby$$

where α and β are first order formulas, such that a, b and y do not appear free in α ; a, b and x do not appear free in β .

A geometry defined by the elementary continuity axiom schema instead of the higher order continuity axiom is called elementary.

12 Reflexivity of β

$$\beta ABB$$

B is always between A and B .

14 Symmetry of β

$$\beta ABC \Rightarrow \beta CBA$$

If B is between A and C then B is between C and A .

13 Compatibility of equality with β

$$A = B \Rightarrow \beta ABA$$

19 Compatibility of equality with \equiv

$$A = B \Rightarrow AC \equiv BC$$

⁴ $ColABC$ is defined by $\beta ABC \vee \beta BCA \vee \beta CAB$

15 Transitivity (inner) of β

$$\beta ABD \wedge \beta BCD \Rightarrow \beta ABC$$

16 Transitivity (outer) of β

$$\beta ABC \wedge \beta BCD \wedge B \neq C \Rightarrow \beta ABD$$

17 Connectivity (inner) of β

$$\beta ABD \wedge \beta ACD \Rightarrow \beta ABC \vee \beta ACB$$

18 Connectivity (outer) of β

$$\beta ABC \wedge \beta ABD \wedge A \neq B \Rightarrow \beta ACD \vee \beta ADC$$

20 Triangle construction unicity

$$\begin{aligned} AC \equiv AC' \wedge BC \equiv BC' \wedge \\ \beta ADB \wedge \beta AD'B \wedge \beta CDX \wedge \Rightarrow C = C' \\ \beta C'D'X \wedge D \neq X \wedge D' \neq X \end{aligned}$$

20₁ Triangle construction unicity (variant)

$$\begin{aligned} A \neq B \wedge \\ AC \equiv AC' \wedge BC \equiv BC' \wedge \quad \Rightarrow C = C' \\ \beta BDC' \wedge (\beta ADC \vee \beta ACD) \end{aligned}$$

21 Triangle construction existence

$$AB \equiv A'B' \Rightarrow \exists CX, AC \equiv A'C' \wedge BC \equiv B'C' \wedge \beta CXP \wedge (\beta ABX \vee \beta BXA \vee \beta XAB)$$

3.2 History

Tarski began to work on his axiom system in 1926 and presented it during his lectures at Warsaw university⁵. He submitted it for publication in 1940 and was first published in his first form in 1967 [Tar67]. This version contains 20 axioms and one schema. A second version, a bit simpler was published in [Tar51]. This first simplification consist only in considering a logic with built-in equality, axioms 13 and 19 are then useless. This second version was further simplified by Eva Kallin, Scott Taylor and Tarski into a system of twelve axioms [Tar59]. The last simplification was obtained by Gupta in its thesis [Gup65], he gives the proof that two more axioms can be derived from the remaining ones.

Figure 3 gives the list of axioms contained in each of these axiom systems. Figure 4 provides the final list of axioms that we used in our formalization.

⁵ We use [TG99] and the footnotes in [Tar51] to give a quick history of the different versions of Tarski's axiom system.

Year :	1940	1951	1959	1965	1983
Reference :	[Tar67]	[Tar51]	[Tar59]	[Gup65]	[SST83]
Axioms :	1	1	1	1	1
	2	2	2	2	2
	3	3	3	3	3
	4	4	4	4	4
	5 ₁	5 ₁	5	5	5
	6	6	6		6
	7 ₂	7 ₂	7 ₁	7 ₁	7
	8(2)	8(2)	8(2)	8(2)	8(2)
	9 ₁ (2)	9 ₁ (2)	9(2)	9(2)	9(2)
	10	10	10 ₁	10 ₁	10
	11	11	11	11	11
	12	12			
	13				
	14	14			
	15	15	15	15	
	16	16			
	17	17			
	18	18	18		
	19				
	20	→ 20 ₁			
	21	21			
Nb of axioms :	20	18	12	10	10
	+	+	+	+	+
	1 schema	1 schema	1 schema	1 schema	1 schema

Fig. 3. History of Tarski's axiom systems.

Identity $\beta ABA \Rightarrow (A = B)$
 Pseudo-Transitivity $AB \equiv CD \wedge AB \equiv EF \Rightarrow CD \equiv EF$
 Reflexivity $AB \equiv BA$
 Identity $AB \equiv CC \Rightarrow A = B$
 Pasch $\exists X, \beta APC \wedge \beta BQC \Rightarrow \beta PxB \wedge \beta QxA$
 Euclid $\exists XY, \beta ADT \wedge \beta BDC \wedge A \neq D \Rightarrow$
 $\beta PxB \wedge \beta QxA$
 $AB \equiv A'B' \wedge BC \equiv B'C' \wedge$
 5 segments $AD \equiv A'D' \wedge BD \equiv B'D' \wedge$
 $\beta ABC \wedge \beta A'B'C' \wedge A \neq B \Rightarrow CD \equiv C'D'$
 Construction $\exists E, \beta ABE \wedge BE \equiv CD$
 Lower Dimension $\exists ABC, \neg \beta ABC \wedge \neg \beta BCA \wedge \neg \beta CAB$
 Upper Dimension $AP \equiv AQ \wedge BP \equiv BQ \wedge CP \equiv CQ \wedge P \neq Q$
 $\Rightarrow \beta ABC \vee \beta BCA \vee \beta CAB$
 Continuity $\forall XY, (\exists A, (\forall xy, x \in X \wedge y \in Y \Rightarrow \beta Axy)) \Rightarrow$
 $\exists B, (\forall xy, x \in X \Rightarrow y \in Y \Rightarrow \beta xBy).$

Fig. 4. Tarski's axiom system (Formalized version - 11 axioms).

4 Formalization in Coq

The mechanization of the proof we have realized prove formally that the simplifications of the first version of Tarski's axiom system are correct. The unnecessary axioms are derived from the remaining ones.

Now, we provide a quick overview of the content of each chapter. We will only detail an example proof in the next section.

The first chapter contains the axioms and the definition of the collinearity predicate (noted `Col`).

The second chapter contains some basic properties of the equidistance predicate (noted `Cong`). It contains also the proof of the unicity of the point constructed thanks to the segment construction axiom.

The third chapter contains some properties of the betweenness predicate (noted `Bet`). It contains in particular the proof of the axioms 12, 14 and 16.

The fourth chapter contains the proof of several properties of `Cong`, `Col` and `Bet`.

The fifth chapter contains some pseudo-transitivity properties of betweenness and the definition of the length comparison predicate (noted `le`) with some associated properties. It includes in particular the proofs of the axioms 17 and 18.

The sixth chapter defines the `out` predicate which means that a point lies on a line out of a segment. This predicate is used to prove some other properties of `Cong`, `Col` and `Bet` such as transitivity properties for `Col`.

The seventh chapter defines the midpoint of a segment and symmetric points. It has to be noted that at this step the existence of the midpoint is not derived yet.

The eighth chapter contains the definition of the perpendicular predicate (noted `Perp`), and the proof of some related properties such as the existence of the foot of the perpendicular. Finally, the existence of the midpoint of a segment is derived.


4.1 Two crucial lemmas

Our formalization follows strictly the lines of the book by Schwabhäuser, Szmielew and Tarski except in the fifth chapter where we introduce two crucial lemmas which do not appear in the original text. These two lemmas allows to deduce the equality of two points which lie on a segment under an hypotheses involving distances.

$$\forall ABC, \beta ABC \wedge AC \equiv AB \Rightarrow C = B$$



$$\forall ABDE, \beta ADB \wedge \beta AEB \wedge AD \equiv AE \Rightarrow D = E.$$



4.2 A comparison between the formal and informal proofs

We reproduce here one of the non trivial proofs: the proof due to Gupta [Gup65] that axiom 18 can be derived from the remaining ones. We translate the proof from [SST83] and provide in parallel the mechanized proof as a Coq script.

For the reader not familiar with the Coq proof assistant, we provide a quick informal explanation of the role of the main tactics we use in this proof.

- `assert` is used to state what we want to prove. When it is followed by “...” this means that this assertion can be proved automatically.
- `DecompExAnd`, given an existential hypotheses, introduces the witness of the existential and decompose the knowledge about it.
- `apply` is used to apply a lemma or theorem.
- `Tarski,sTarski,Between,...` are automatic tactics which try to prove the current goal. Informally this can be read as “by simple properties of betweenness” or “by direct application of one of the axioms”.
- `unfold` replaces something by its definition.
- `cases_equality` perform a reasoning by cases on the equality of two points.




Fig. 5. Proof of axiom 18

Theorem 1 (Gupta). $A \neq B \wedge \beta ABC \wedge \beta ABD \Rightarrow \beta ACD \vee \beta ADC$

Preuve: Let C' and D' be points such that :

$$\beta ADC' \wedge DC' \equiv CD \text{ and } \beta AC'D' \wedge CD' \equiv CD$$

```

assert (exists C', Bet A D C' /\ Cong D C' C D)...
DecompExAnd H2 C'.
assert (exists D', Bet A C D' /\ Cong C D' C D)...
DecompExAnd H2 D'.

```

*We have to show that $C = C'$ or $D = D'$.
Let B and B'' points such that :*

$$\beta AC' B' \wedge C' B' \equiv CB \text{ and } \beta AD' B'' \wedge D' B'' \equiv DB$$

```

assert (exists B', Bet A C' B' /\ Cong C' B' C B)...
DecompExAnd H2 B'.
assert (exists B'', Bet A D' B'' /\ Cong D' B'' D B)...
DecompExAnd H2 B''.

```

Using the lemma 2.11⁶ we can deduce that $BC' \equiv B''C$ and that $BB' \equiv B''B$.

```

assert (Cong B C' B'' C).
eapply l2_11.
3:apply cong_commutativity.
3:apply cong_symmetry.
3:apply H11.
Between.
Between.
esTarski.
assert (Cong B B' B'' B).
eapply l2_11;try apply H2;Between.

```

By unicity of the segment construction, we know that $B'' = B'$.

```

assert (B''=B').
apply construction_unicity with
(Q:=A) (A:=B) (B:=B'') (C:=B) (x:=B'') (y:=B');Between...
smart_subst B''.

```

We know that $FSC \left(\begin{smallmatrix} BCD'C' \\ B'C'DC \end{smallmatrix} \right)$ (The points form a five segments configuration).

```

assert (FSC B C D' C' B' C' D C).
unfold FSC;repeat split;unfold Col;Between;sTarski.
2:eapply cong_transitivity.
2:apply H7.
2:sTarski.
apply l2_11 with (A:=B) (B:=C) (C:=D') (A':=B') (B':=C') (C':=D);
Between;sTarski;esTarski.

```

Hence $C'D' \equiv CD$ (because if $B \neq C$ the five segments axiom gives the conclusion and if $B = C$ we can use the hypotheses).

⁶ The lemma 2.11 states that $\beta ABC \wedge \beta A'B'C' \wedge AB \equiv A'B' \wedge BC \equiv B'C' \Rightarrow AC \equiv A'C'$.

```

assert (Cong C' D' C D).
cases_equality B C.
(* First case *)
treat_equalities.
eapply cong_transitivity.
apply cong_commutativity.
apply H11.
Tarski.
(* Second case *)
apply cong_commutativity.
eapply l4_16;try apply H3...

```

Using the axiom of Pasch, there is a point E such that :

$$\beta C E C' \wedge \beta D E D'$$

```

assert (exists E, Bet C E C' /\ Bet D E D').
eapply inner_pash;Between.
DecompExAnd H13 E.

```

We can deduce that IFS $\left(\begin{smallmatrix} ded'c \\ ded'c' \end{smallmatrix}\right)$ and IFS $\left(\begin{smallmatrix} cec'd \\ cec'd' \end{smallmatrix}\right)$.

```

assert (IFSC D E D' C D E D' C').
unfold IFSC;repeat split;Between;sTarski.
eapply cong_transitivity.
apply cong_commutativity.
apply H7.
sTarski.

```

```

assert (IFSC C E C' D C E C' D').
unfold IFSC;repeat split;Between;sTarski.
eapply cong_transitivity.
apply cong_commutativity.
apply H5.
sTarski.

```

Hence $EC \equiv EC'$ and $ED \equiv ED'$.

```

assert (Cong E C E C').
eapply l4_2;eauto.
assert (Cong E D E D').
eapply l4_2;eauto.

```

Suppose that $C \neq C'$. We have to show that $D = D'$ ⁷.

⁷ Note that this step uses the decidability of equality between two points.

```

cases_equality C C'.
smart_subst C'.
assert (E=C).
eTarski.
smart_subst E.
unfold IFSC, FSC, Cong_3 in *;intuition.

```

From the hypotheses, we can infer that $C \neq D'$.

```

assert (C<>D').
unfold not;intro.
treat_equalities...

```

Using the segment construction axiom, we know that there are points P , Q and R such that :

$$\beta C' C P \wedge CP \equiv CD' \text{ and } \beta D' C R \wedge CR \equiv CE \text{ and } \beta P R Q \wedge RQ \equiv RP$$

```

assert (exists P, Bet C' C P /\ Cong C P C D')...
DecompExAnd H21 P.
assert (exists R, Bet D' C R /\ Cong C R C E)...
DecompExAnd H21 R.
assert (exists Q, Bet P R Q /\ Cong R Q R P)...
DecompExAnd H21 Q.

```

Hence $FSC \left(\begin{smallmatrix} D'CRP \\ PCED' \end{smallmatrix} \right)$, so $RP \equiv ED'$ and $RQ \equiv ED$.

```

assert (FSC D' C R P P C E D').
unfold FSC;unfold Cong_3;intuition...
eapply l2_11.
apply H25.
3:apply H26.
Between.
sTarski.

```

```

assert (Cong R P E D').
eapply l4_16.
apply H21.
auto.

```

```

assert (Cong R Q E D).
eapply cong_transitivity.
apply H28.
eapply cong_transitivity.
apply H22.
sTarski.

```

We can infer that $FSC \left(\begin{smallmatrix} D'EDC \\ PRQC \end{smallmatrix} \right)$,

```

assert (FSC D' E D C P R Q C).
unfold FSC;unfold Cong_3;intuition...
eapply l2_11.
3:eapply cong_commutativity.
3:eapply cong_symmetry.
3:apply H22.
Between.
Between.
sTarski.

```

so using lemma 2.11 we can conclude that $D'D \equiv PQ$ and $CQ \equiv CD$ (because the case $D' \neq E$ is solved using the five segments axiom, and in the other case we can deduce that $D' = D$ and $P = Q$).

```

cases_equality D' E.
(* First case *)
treat_equalities...
sTarski.
(* Second case *)
eapply l4_16;eauto.

```

Using the theorem 4.17⁸, as $R \neq C$ and R, C and D' are collinear we can conclude that $D'P \equiv D'Q$.

```

assert (R<>C).
unfold not;intro.
treat_equalities...

assert (Cong D' P D' Q).
apply l4_17 with (A:=R) (B:=C) (C:=D').
assumption.
3:apply H32.
unfold Col;left;Between.
sTarski.

```

As $C \neq D'$, $Col CD'B$ and $Col CD'B'$, we can also deduce that $BP \equiv BQ$ and $B'P \equiv B'Q$.

```

assert (Cong B P B Q).
eapply l4_17; try apply H20;auto.
unfold Col;right;right;Between.
(* *)
assert (Cong B' P B' Q).
eapply l4_17 with (C:=B').
apply H20.

```

⁸ The theorem 4.17 states that $A \neq B \wedge ColABC \wedge AP \equiv AQ \wedge BP \equiv BQ \Rightarrow CP \equiv CQ$.

```

unfold Col.
Between.
assumption.
assumption.

```

As $C \neq D'$, we have $B \neq B'$ and as $Col BC'B'$ we have $C'P \equiv C'Q$.

```

cases_equality B B'.
subst B'.
unfold IFSC,FSC, Cong_3 in *;intuition.
clean_duplicated_hyps.
clean_trivial_hyps.
assert (Bet A B D').
Between.
assert (B=D').
eTarski.
treat_equalities.
Tarski.

```

```

assert (Cong C' P C' Q).
eapply l4_17.
apply H37.
unfold Col;right;left;Between.
auto.
auto.

```

As $C \neq C'$ and $Col C'CP$ we have $PP \equiv PQ$.

```

assert (Cong P P P Q).
eapply l4_17.
apply H19.
unfold Col;right;right;Between.
auto.
auto.

```

Using the identity axiom for equidistance, we can deduce that $P = Q$.

```

assert (P=Q).
eapply cong_identity.
apply cong_symmetry.
apply H39.

```

As $PQ \equiv D'D$, we also have $D = D'$.

```

subst Q.
assert (D=D').
eapply cong_identity with (A:=D) (B:=D') (C:=P).
unfold IFSC,FSC, Cong_3 in *;intuition.

```

The proof is finished.

```
assert (E=D).
eTarski.
unfold IFSC,FSC, Cong_3 in *;intuition.
```

□

4.3 About degenerated cases

Every paper about the formalization of geometry, in particular those about Hilbert's foundations of geometry [DDS00,MF03] emphasizes the problem of the degenerated cases. The degenerated cases are limit cases such as when two points are equals, three points are collinear or two lines are parallel. The formal proof of the theorems in the degenerated cases is often tedious and even sometimes difficult. These cases often do not even appear in the informal proof⁹. In order to limit the size of the proofs, we tried to automate some tasks. These pieces of automation should not be compared with the highly successful decision procedures for geometry, the goal is just to automate some easy but very tedious proofs and as our goal is to build foundations for the implementation of decision procedures we can not use these more powerful procedures.

The main tactic to deal with degenerated cases is called `treat_equalities`. The basic idea is to propagate information about degenerated cases. For instance, if we know that $A = B$ and $AB \equiv CD$ we can deduce that $C = D$. This is very simple but it shortens the proofs of the degenerated cases quite effectively.

Moreover, we think that a source of degenerated cases come from the axiom system. In our personal experience the formalization of geometry using Hilbert axioms lead to far more degenerated cases because the axioms are not always stated in the most general and uniform way. We think that Tarski's geometry is a good candidate to mechanization because it is very simple, it has good meta-mathematical properties (cf [Tar51]) and it produces few degenerated cases.

4.4 Classical vs intuitionist logic.

Our formalization of Tarski's geometry is performed in the system Coq. As the logic behind Coq is constructive, we need to tell Coq explicitly when we need classical logic. This is the case in this development. It appears quite often in the proofs that we need to distinguish between two cases such that $A = B$ and $A \neq B$ or $ColABC$ and $\neg ColABC$. This kind of reasoning relies on the decidability of point equality and collinearity. We proved these two facts using the excluded middle rule.

⁹ It seems that degenerated cases play the same role in geometry as α -conversion in lambda calculus: they are a great source of difficulties in the context of a mechanization.

5 Future work

A natural extension of our work consist in mechanizing the remaining chapters of [SST83] and proving the axioms of Hilbert. This work is under progress. We also plan to enrich our formalization to use it as a foundation for other formal Coq developments about geometry such as Frédérique Guilhot formalization of geometry as it is presented in the french curriculum [Gui05] and our implementation in Coq of the area method of Chou, Gao and Zhang [Nar04]. A longer-term challenge would be to perform a systematic development of geometry similar to the book of Schwabhäuser, Szmielew and Tarski but in the context of a constructive axiom system such as the axiom system of von Plato [vP95] which has already been formalized in the Coq proof assistant by Gilles Khan [Kah95]

6 Conclusion

We have presented the mechanisation of the proofs of over 150 lemmas in the context of Tarski's geometry. This includes the formal proof that the simplifications of the first version of Tarski's axiom system are corrects. Our main conclusion is that Tarski axiom system lead to more uniform proofs than Hilbert's axiom system and so it is better suited for a formalization.

Availability

The full Coq development with the formal *proofs* and hypertext links to ease navigation can be found at the following url :

<http://www.lix.polytechnique.fr/Labo/Julien.Narboux/tarski.html>

References

- [CGZ94] Shang-Ching Chou, Xiao-Shan Gao, and Jing-Zhong Zhang. *Machine Proofs in Geometry*. World Scientific, Singapore, 1994.
- [Coq04] Coq development team, The. *The Coq proof assistant reference manual, Version 8.0*. LogiCal Project, 2004.
- [DDS00] Christophe Dehlinger, Jean-François Dufourd, and Pascal Schreck. Higher-order intuitionistic formalization and proofs in Hilbert's elementary geometry. In *Automated Deduction in Geometry*, pages 306–324, 2000.
- [Gui05] Frédérique Guilhot. Formalisation en Coq et visualisation d'un cours de géométrie pour le lycée. *Revue des Sciences et Technologies de l'Information, Technique et Science Informatiques, Langages applicatifs*, 24:1113–1138, 2005. Lavoisier.
- [Gup65] Haragauri Narayan Gupta. *Contributions to the axiomatic foundations of geometry*. PhD thesis, University of California, Berkley, 1965.
- [Kah95] Gilles Kahn. Constructive geometry according to Jan von Plato. Coq contribution, 1995. Coq V5.10.

- [MF03] Laura Meikle and Jacques Fleuriot. Formalizing Hilbert's Grundlagen in Isabelle/Isar. In *Theorem Proving in Higher Order Logics*, pages 319–334, 2003.
- [MF05] Laura Meikle and Jacques Fleuriot. Mechanical theorem proving in computation geometry. In Hoon Hong and Dongming Wang, editors, *Automated Deduction in Geometry 2004*, volume 3763 of *Lecture Notes in Computer Science*, pages 1–18. Springer-Verlag, November 2005.
- [Nar04] Julien Narboux. A decision procedure for geometry in Coq. In Slind Konrad, Bunker Annett, and Gopalakrishnan Ganesh, editors, *Proceedings of TPHOLS'2004*, volume 3223 of *Lecture Notes in Computer Science*. Springer-Verlag, 2004.
- [Nar05] Julien Narboux. Toward the use of a proof assistant to teach mathematics. In *Proceedings of the 7th International Conference on Technology in Mathematics Teaching (ICTMT7)*, 2005.
- [Pau] Lawrence C. Paulson. The Isabelle reference manual.
- [PB01] David Pichardie and Yves Bertot. Formalizing convex hulls algorithms. In *Proc. of 14th International Conference on Theorem Proving in Higher Order Logics (TPHOLS'01)*, volume 2152 of *Lecture Notes in Computer Science*, pages 346–361. Springer-Verlag, 2001.
- [Qua89] Art Quaife. Automated development of tarski's geometry. *Journal of Automated Reasoning*, 5(1):97–118, 1989.
- [SST83] Wolfram Schwabhäuser, Wanda Szmielew, and Alfred Tarski. *Metamathematische Methoden in der Geometrie*. Springer-Verlag, Berlin, 1983.
- [Tar51] Alfred Tarski. *A decision method for elementary algebra and geometry*. University of California Press, 1951.
- [Tar59] Alfred Tarski. What is elementary geometry? In P. Suppes L. Henkin and A. Tarski, editors, *The axiomatic Method, with special reference to Geometry and Physics*, pages 16–29, Amsterdam, 1959. North-Holland.
- [Tar67] Alfred Tarski. The completeness of elementary algebra and geometry, 1967.
- [TG99] Alfred Tarski and Steven Givant. Tarski's system of geometry. *The bulletin of Symbolic Logic*, 5(2), June 1999.
- [vP95] Jan von Plato. The axioms of constructive geometry. In *Annals of Pure and Applied Logic*, volume 76, pages 169–200, 1995.